


PUT and INPUT functions: Making more use of formats and informats

James Zeitler
Baker Research Services
Harvard Business School
jzeitler@hbs.edu

INPUT Statement and INFORMAT



Reading Text into SAS: INPUT Statement and INFORMAT

		Ticker	Prc	Vol
12490	20070131	IBM	99.15000	1928647
12490	20070228	IBM	92.94000	1257950
12490	20070330	IBM	94.26000	1634895
12490	20070430	IBM	102.21000	1668675

```
data work.PriceVolume;
infile 'c:\data\MonthlyPriceVolume.txt' firstobs = 2;
informat PermID 5.
        Date yymmdd8.
        Ticker $4.
        Price 12.
        Volume 12.;
input PermID Date Ticker Price Volume;
run;
```

March 28, 2008

BASUG

2

Writing Text out of SAS

```
Title 'Month-End Closing Price and Total Volume';
run;
proc print data = work.PriceVolume;
run;
```

Month-End Closing Price and Total Volume					
Obs	Perm ID	Date	Ticker	Price	Volume
1	12490	17197	IBM	99.150	1928647
2	12490	17225	IBM	92.940	1257950
3	12490	17255	IBM	94.260	1634895
4	12490	17286	IBM	102.210	1668675

March 28, 2008

BASUG

3

Writing Text out of SAS: FORMAT

```
proc print data = work.PriceVolume;
  format date date9.;
run;
```

Month-End Closing Price and Total Volume


Obs	Perm ID	Date	Ticker	Price	Volume
1	12490	31JAN2007	IBM	99.150	1928647
2	12490	28FEB2007	IBM	92.940	1257950
3	12490	30MAR2007	IBM	94.260	1634895
4	12490	30APR2007	IBM	102.210	1668675
5	12490	31MAY2007	IBM	106.600	1466071

March 28, 2008

BASUG

4

PUT Statement and FORMAT



March 28, 2008

BASUG

5

Writing Text out of SAS: PUT Statement and FORMAT

```
data _null_;
  set work.PriceVolume (where = (Ticker = "GOOG"));
  format Date date9.;
  put Ticker $4. +1 Date +1 Price 12.5 +1 Volume 12.0;
run;
```

GOOG	31JAN2007	501.50000	1251505
GOOG	28FEB2007	449.45001	1142896
GOOG	30MAR2007	458.16000	1183525
GOOG	30APR2007	471.38000	934070
GOOG	31MAY2007	497.90900	1010372
GOOG	29JUN2007	522.70007	1294012


PUT writes text, strings of characters, that can be interpreted outside of SAS as words, labels, dates, numbers, etc.

March 28, 2008

BASUG

6

INPUT and PUT Functions: Translating Data within SAS



March 28, 2008

BASUG

7

Creating a New Variable with the PUT Function and a FORMAT

```

title "Daily Price and Volume";
run;
data work.DailyPriceVolume;
  infile 'c:\data\DailyPriceVolume.txt' firstobs = 2;
  informat PermID 5.
           Date yymmdd8.
           Ticker $4.
           Price 12.
           Volume 12.;
  input PermID Date Ticker Price Volume;
  format Date date9.;
  yyyyymmnm = year(Date)*100 + Month(Date);
  yyyyymmnc = put(Date, yymnm6.);
run;
proc print data = work.DailyPriceVolume (where = (Ticker = "GOOG"));
run;

```

Daily Price and Volume							
Obs	Perm ID	Date	Ticker	Price	Volume	yyyyymmnm	yyyyymmnc
503	90319	03JAN2007	GOOG	467.590	7725290	200701	200701
504	90319	04JAN2007	GOOG	483.260	7944390	200701	200701

March 28, 2008

BASUG

8

PUT Function

- Creates a new variable by applying a **format** to translate the contents of a source variable
- `var2 = PUT(var1, format.);`
 - *format* must match type of source variable, var1
 - Numeric format for numeric source (var1) variable
 - Character format for character source (var1) variable
 - Output variable from PUT function, var2, is **always character**
 - Just as the PUT statement always write text

March 28, 2008

BASUG

9

INPUT Function

- Creates a new variable by applying an **informat** to translate the contents of a source variable
- `var2 = INPUT(source, informat.);`
 - *Source* should be text or character variable
 - INPUT reads text
 - *Informat* translates text to internal SAS storage format
 - Output variable from INPUT function, *var2*, is character or numeric, depending on the INFORMAT
 - Numeric *informat* creates numeric variable
 - Character *informat* creates character variable

March 28, 2008

BASUG

10

Using INPUT Function

```
yyymmcc = put(Date, yymmn6.);
yyymm1 = input(yyymmcc, 6.0);
```

- **put(Date, yymmn6.)**
 - Translates date into a character variable **yyymmcc** using the **yymmn6. format**
- **input(yyymmcc, 6.0)**
 - Translates character variable **yyymmcc** into a numeric variable using the **6.0 informat**

March 28, 2008

BASUG

11

Using PUT and INPUT Functions in Tandem

```
yyyymn1 = input(put(date, yymmn6.), 6.0);
```

- **put(date, yymmn6.)**
 - Translates date into a 6-character text string using the yymmn6. format
- **input(put(date, yymmn6.), 6.0)**
 - Translates *put(date, yymmn6.)* into a SAS numeric variable using the 6.0 informat

March 28, 2008

BASUG

12

A Couple More Examples

Price by Month by Year

Obs	Ticker	Jan_2007	Feb_2007	Mar_2007
1	IBM	99.15	92.940	94.26
2	MSFT	30.86	28.170	27.87
3	GOOG	501.50	449.450	458.16

```
proc sort data = work.PriceByMonth (where = (ticker gt ''));
  by ticker;
run;
proc transpose data = work.PriceByMonth
  out = work.PriceByMonthT (rename = (Col1 = Price));
  by ticker;
run;
data work.PriceByMonthT (drop = _label_ _name_);
  set work.PriceByMonthT;
  Date = input(put(_name_, $8.0), monyy8.);
  format Date date9.;
run;
```

March 28, 2008

BASUG

13

The Translated Variable Names

Price by Month by Year			
Obs	Ticker	Price	Date
1	GOOG	501.50	01JAN2007
2	GOOG	449.45	01FEB2007
3	GOOG	458.16	01MAR2007
4	IBM	99.15	01JAN2007
5	IBM	92.94	01FEB2007
6	IBM	94.26	01MAR2007
7	MSFT	30.86	01JAN2007
8	MSFT	28.17	01FEB2007
9	MSFT	27.87	01MAR2007

March 28, 2008

BASUG

14

Using User-Defined INFORMAT and FORMAT

```

proc format;
  invalue TicIDf "GOOG" = 90319
                "IBM"  = 12490
                "MSFT" = 10107;
  value  $TicNamF "GOOG" = "GOOGLE INC"
                "IBM"  = "INTERNATIONAL BUSINESS MACHINES"
                "MSFT" = "MICROSOFT CORP";
run;
data work.PriceByMonthT;
  set work.PriceByMonthT;
  CoName = put(ticker,$TicNamf.);
  PermID = input(ticker,TicIDf.);
run;

```

March 28, 2008

BASUG

15

Name and ID

Price by Month by Year					
Obs	Ticker	Price	Date	CoName	Perm ID
1	GOOG	501.50	01JAN2007	GOOGLE INC	90319
2	GOOG	449.45	01FEB2007	GOOGLE INC	90319
3	GOOG	458.16	01MAR2007	GOOGLE INC	90319
4	IBM	99.15	01JAN2007	INTERNATIONAL BUSINESS MACHINES	12490
5	IBM	92.94	01FEB2007	INTERNATIONAL BUSINESS MACHINES	12490
6	IBM	94.26	01MAR2007	INTERNATIONAL BUSINESS MACHINES	12490
7	MSFT	30.86	01JAN2007	MICROSOFT CORP	10107
8	MSFT	28.17	01FEB2007	MICROSOFT CORP	10107
9	MSFT	27.87	01MAR2007	MICROSOFT CORP	10107

March 28, 2008

BASUG

16